title: Could not resolve com.squareup.sqldelight:runtime:1.1.3
body: <p>I tried to integrate sqldelight in my Multiplatform library project for Android/iOS but I'm having several unresolved dependency errors when syncing gradle.</p>
  
  <blockquote>
    <p>ERROR: Unable to resolve dependency for ':SharedCode@debug/compileClasspath': Could not resolve com.squareup.sqldelight:runtime:1.1.3.</p>
    
    <p>ERROR: Unable to resolve dependency for ':SharedCode@debug/compileClasspath': Could not resolve com.squareup.sqldelight:runtime-jvm:1.1.3.</p>
    
    <p>ERROR: Unable to resolve dependency for ':SharedCode@release/compileClasspath': Could not resolve com.squareup.sqldelight:runtime:1.1.3.
    ERROR: Unable to resolve dependency for ':SharedCode@release/compileClasspath': Could not resolve com.squareup.sqldelight:runtime-jvm:1.1.3.**</p>
  </blockquote>
  
  <ul>
  <li><p>Gradle version 5.1.1</p></li>
  <li><p>Gradle Plugin 3.4.0</p></li>
  <li><p>sqldelight 1.1.3</p></li>
  </ul>
  
  <p><code>enableFeaturePreview('GRADLE_METADATA')</code> is present in my settings.gradle</p>
  
  <p>My project gradle file looks like this:</p>
  
  <pre><code>buildscript {
      ext {
          kotlin_version = '1.3.31'
          ktor_version = '1.2.1'
          ktor_json_version = '1.2.1'
          kotlinx_coroutines_version = '1.2.1'
          serialization_version = '0.11.0'
          sqldelight_version = '1.1.3'
          dokka_version = '0.9.16'
      }
  
      repositories {
          google()
          jcenter()
          mavenCentral()
          gradlePluginPortal()
      }
      dependencies {
          classpath 'com.android.tools.build:gradle:3.4.0'
          classpath "org.jetbrains.kotlin:kotlin-gradle-plugin:$kotlin_version"
          classpath "org.jetbrains.kotlin:kotlin-serialization:$kotlin_version"
          classpath "org.jetbrains.dokka:dokka-android-gradle-plugin:$dokka_version"
          classpath "com.squareup.sqldelight:gradle-plugin:$sqldelight_version"
      }
  }
  
  allprojects {
      repositories {
          google()
          jcenter()
          mavenCentral()
          maven { url "https://kotlin.bintray.com/kotlinx" }
      }
  }
  
  task clean(type: Delete) {
      setDelete rootProject.buildDir
  }
  </code></pre>
  
  <p>my SharedCode lib gradle file:</p>
  
  <pre><code>apply plugin: 'kotlinx-serialization'
  apply plugin: 'com.android.library'
  apply plugin: 'org.jetbrains.kotlin.multiplatform'
  apply plugin: 'com.squareup.sqldelight'
  
  group = 'com.example.multiplatform'
  version = '1.0'
  
  android {
      compileSdkVersion 27
      defaultConfig {
          minSdkVersion 15
      }
      buildTypes {
          release {
              minifyEnabled false
          }
      }
  }
  
  dependencies {
      // Specify Kotlin/JVM stdlib dependency.
      implementation 'org.jetbrains.kotlin:kotlin-stdlib-jdk7'
      implementation "com.squareup.sqldelight:runtime:1.1.3"
  
      testImplementation 'junit:junit:4.12'
      testImplementation 'org.jetbrains.kotlin:kotlin-test'
      testImplementation 'org.jetbrains.kotlin:kotlin-test-junit'
  
      androidTestImplementation 'junit:junit:4.12'
      androidTestImplementation 'org.jetbrains.kotlin:kotlin-test'
      androidTestImplementation 'org.jetbrains.kotlin:kotlin-test-junit'
  
      androidTestImplementation 'com.android.support.test:runner:1.0.2'
      androidTestImplementation 'com.android.support.test.espresso:espresso-core:3.0.2'
  }
  
  kotlin {
      targets {
          final def iOSTarget = System.getenv('SDK_NAME')?.startsWith("iphoneos")  \
                             ? presets.iosArm64 : presets.iosX64
  
          fromPreset(iOSTarget, 'ios') {
              binaries {
                  framework('SharedCode')
              }
          }
  
          fromPreset(presets.android, 'androidLib')
      }
  
      sourceSets {
          commonMain {
              dependencies {
                  //HTTP
                  implementation "io.ktor:ktor-client-json:$ktor_json_version"
                  implementation "io.ktor:ktor-client-serialization:$ktor_version"
                  //Coroutines
                  implementation "org.jetbrains.kotlinx:kotlinx-coroutines-core-common:$kotlinx_coroutines_version"
                  //Kotlinx serialization
                  implementation "org.jetbrains.kotlinx:kotlinx-serialization-runtime-common:$serialization_version"
                  //sqldelight
                  implementation "com.squareup.sqldelight:runtime:$sqldelight_version"
              }
          }
  
          androidLibMain {
              dependencies {
                  //HTTP
                  implementation "io.ktor:ktor-client-json-jvm:$ktor_json_version"
                  implementation "io.ktor:ktor-client-serialization-jvm:$ktor_version"
                  //Coroutines
                  implementation "org.jetbrains.kotlinx:kotlinx-coroutines-android:$kotlinx_coroutines_version"
                  //Kotlinx serialization
                  implementation "org.jetbrains.kotlinx:kotlinx-serialization-runtime:$serialization_version"
                  //sqldelight
                  implementation "com.squareup.sqldelight:android-driver:$sqldelight_version"
              }
          }
  
          iosMain {
              dependencies {
                  //HTTP
                  implementation "io.ktor:ktor-client-ios:$ktor_version"
                  implementation "io.ktor:ktor-client-json-native:$ktor_json_version"
                  implementation "io.ktor:ktor-client-serialization-native:$ktor_version"
                  //Coroutines
                  implementation "org.jetbrains.kotlinx:kotlinx-coroutines-core-native:$kotlinx_coroutines_version"
                  //kotlinx serialization
                  implementation "org.jetbrains.kotlinx:kotlinx-serialization-runtime-native:$serialization_version"
                  //sqldelight
                  implementation "com.squareup.sqldelight:ios-driver:$sqldelight_version"
              }
          }
      }
  }
  
  sqldelight {
      MyApp {
          packageName = 'com.example.multiplatform'
      }
  }
  
  configurations {
      compileClasspath
  }
  
  task packForXCode(type: Sync) {
      final File frameworkDir = new File(buildDir, "xcode-frameworks")
      final String mode = project.findProperty("XCODE_CONFIGURATION")?.toUpperCase() ?: 'DEBUG'
      final def framework = kotlin.targets.ios.binaries.getFramework("SharedCode", mode)
  
      inputs.property "mode", mode
      dependsOn framework.linkTask
  
      from { framework.outputFile.parentFile }
      into frameworkDir
  
      doLast {
          new File(frameworkDir, 'gradlew').with {
              text = "#!/bin/bash\nexport 'JAVA_HOME=${System.getProperty("java.home")}'\ncd '${rootProject.rootDir}'\n./gradlew \$@\n"
              setExecutable(true)
          }
      }
  }
  tasks.build.dependsOn packForXCode
  </code></pre>